
INFORMATIKA

Rozdělení na části

Stanislav Trávníček, PřF UP, Olomouc

Abstract. In the article, we engage in this problem: We have a lot of expensive
things and we need to divide them into three (or several) parts so that all the
parts have the same value or the nearest value. The problem is solved by a
computer program in two ways.

V článku [1] řešil pan Koumal problém, jak odnést nákup ve dvou
taškách tak, aby byly stejně těžké (tedy aby rozdíl jejich hmotností byl
co nejmenší), a vyřešil ho úspěšně užitím programu v článku uvedeném.
Může nás samozřejmě napadnout, že by se nějak podobně měla řešit
úloha pro rozdělení celku na k > 2 stejných dílů; uvážíme případ k = 3,
protože pro k > 3 se již pak nic nového nedostane.
Pan Koumal nemá ovšem tři ruce, takže si musíme zvolit jinou prak-

tickou interpretaci. Třeba takto: Tři sestry zdědily košík drahých kamenů
(třeba v pohádce) a chtějí si je spravedlivě rozdělit. Každý kámen má
stanovenou cenu, takže za spravedlivé budou považovat takové rozdělení
na tři hromádky, kdy všechny budou mít stejnou hodnotu, nebo kde roz-
díly budou nejmenší možné, a sestry se pak vyrovnají. Proto se rozhodly,
že pověří nestranného člověka, který takové hromádky vytvoří, a ony si
pak každá tu svou vylosují. Tímto úkolem pověřily právě pana Koumala,
protože slyšely, že už něco podobného prováděl.

Obr. 1: Tři sestry (autorkou ilustrace je Jaroslava Palzerová)

Ten si nejprve formuloval kritérium, kdy bude považovat rozdělení za
optimální: když rozdíl hodnoty nejdražší a nejlevnější hromádky bude co
nejmenší (v optimálním případě to může být 0, pak by všechny tři měly
stejnou hodnotu). Tento rozdíl (označený Rozdil) závisí na zadaných
cenách M [K] jednotlivých K = 1 až N prvků. (Samozřejmě to, že jde
o ceny, není vůbec důležité; pro pohodlné srovnání s [1] je v programu
ponechán pojem hmotnost a označení M .)
Pro vyřešení daného problému použijeme kromě způsobu pana Kou-

mala (a už jej dále opusťme) ještě jiný postup, něco jako je metoda Monte
Carlo, a v programu, který je zde prezentován, jsou vždy pro srovnání
uvedeny výsledky obou zpracování; ne vždy se však rovnají a řekneme
si proč.
Některá nastavení v programu nejsou podstatná, například omezení

počtu předmětů číslem MaxN = 15; pro větší N se totiž doba zpracování
viditelně prodlužuje.
Z praktických důvodů je v programu definováno několik procedur.

Procedura Rozloz je základem 1. způsobu zpracování; užitím rekurze
jsou postupně vytvářeny (a uloženy v indexované proměnné A) všechny
N -tice nul, jedniček a dvojek (variace tří prvků s libovolným opaková-
ním). Údaj v K-tém prvku N -tice odpovídá K-tému předmětu – číslům
0, 1, 2 odpovídá zařazení předmětu do 1., 2. nebo 3. skupiny. Tímto způ-
sobem jsou vyčerpány všechny možnosti rozdělení souboru předmětů na
tři části a výsledek takto získaný je naprosto přesný. Pokud se během
práce programu vyhodnotí Rozdil = 0, tak se již v dalším vyšetřování
nepokračuje.
Po vytvoření každé N -tice se provede procedura Vyhodnot . Ta nejprve

zjistí úhrnnou cenu jednotlivých skupin a pak tvoří absolutní hodnoty
rozdílu těchto cen pro zjištění rozdílu mezi nejdražší a nejlevnější sku-
pinou. Poté se testuje, jestli tento největší rozdíl je menší než předchozí
rozdíl (na počátku je zde nastaveno Rozdil := 1 000), a v kladném pří-
padě se uloží k zapamatování nové rozdělení do skupin (TakJeTo), ceny
skupin (S) a nastaví se nová hodnota rozdílu.
Druhý způsob řešení daného problému je založen na náhodných čís-

lech. Je provedena posloupnost, tj. cyklus pokusů (v programu je zvoleno
konstantních 10 000), v každém pokusu se užitím náhodných čísel na-
staví (do A) N -tice nul, jedniček a dvojek a provede se vyhodnocení
procedurou Vyhodnot . Opět se práce ukončí v případě získání nulového
rozdílu.

V obou případech pak následuje procedura Zaver , která vypíše na
obrazovku výsledky uvedených dvou zpracování. Výstup je řešen názorně
a úsporně, aby po konečném zastavení chodu programu byl přehled po
celé práci.
Na obr. 2 je znázorněn výsledek jednoho zpracování; vidíme, že vý-

sledky obou metod jsou věcně stejné, i když rozdělení do skupin je jiné.
Na obr. 3 je zachycena jiná situace.

Obr. 2

Obr. 3

Vidíme, že výsledek použití druhého způsobu výpočtu je horší. Důvod
je zcela prostý. Zatímco prvním způsobem, jak jsme si řekli, jsou pro-
brány všechny možnosti, tak druhým způsobem je probráno sice hodně

možností, ale při generování obsahu A mnoho dalších případů chybělo,
a mezi nimi zřejmě i ty, které by vedly k lepšímu výsledku. Obecně lze
říci, že výsledky jsou tím horší, čím je N větší a čím rozdílnější jsou zadá-
vané hodnoty, a jsou tím lepší, čím relativně (vzhledem k N) více pokusů
se vykoná. Navíc při každém novém zpracování téhož souboru můžeme
dostávat různé výsledky, o čemž se lze jednoduše prakticky přesvědčit
opakovaným prováděním výpočtů.
Výpis programu:

program NaTriX;
uses Crt;
const
MaxN = 15;
BNad = yellow+16*red;
BInf = lightgreen;
BVst = white;
BVys = yellow;
var
N, K, X, Y: Integer;
Rozdil: Real;
R,S,T: array [0..2] of Real;
A, TakJeTo: array [1..MaxN] of Integer;
M: array [1..MaxN] of Real;

procedure Vyhodnot;
var IV,JV: Integer;
begin
for IV := 0 to 2 do T[IV] := 0;
for IV := 1 to N do
T[A[IV]] := T[A[IV]] + M[IV];
for IV := 0 to 2 do
R[IV] := Abs(T[IV]-T[(IV+1) mod 3]);
for IV := 0 to 2 do
begin
if (R[IV]>=R[(IV+1) mod 3]) and (R[IV]>=R[(IV+2) mod 3])
and (R[IV] < Rozdil) then
begin
Rozdil := R[IV];
TakJeTo := A;
S := T
end
end
end; {Vyhodnoť}

procedure Zkusme(J: Integer);
var I, II, L: Integer;
begin
Randomize;
L := 1;
repeat
I := 1;
repeat
A[I] := Random(3);
Inc(I)
until I > J;
Vyhodnot;
Inc(L);
until (L>10000) or (Rozdil = 0)
end; {Zkusme}

procedure Rozloz(J: Integer);
var I, II: Integer;
begin
if J>0 then
begin
A[J] := 0;
Rozloz(J-1);
A[J] := 1;
Rozloz(J-1);
A[J] := 2;
Rozloz(J-1)
end
else
begin
Vyhodnot;
if (Rozdil = 0) then Exit
end
end; {Rozlož}

procedure Zaver(Z: Integer);
var M, L: Integer;
begin
GotoXY(2+35*Z,11);
TextAttr := BInf;
if Z = 0 then Write(’Uvažují se všechny možnosti:’)
else Write(’Výpočet užitím náhodných čísel:’);
TextAttr := BVys;

for M := 0 to 2 do
begin
GotoXY(1+35*Z,13+2*M);
Write(’ Hmotnost ’,M+1,’. skupiny je ’,S[M]:0:3);
GotoXY(1+35*Z,13+2*M+1);
for L := 1 to N do
if TakJeTo[L] = M then Write(’ ’,L);

end;
GotoXY(1+35*Z,13+2*M+2);
Write(’ Rozdíl je ’,Rozdil:0:3)
end; {Zaver}

begin {program}
ClrScr;
WriteLn; TextAttr := BNad;
Write(’Rozdělení skupiny předmětů ’);
WriteLn(’na tři části se stejnou celkovou hmotností.’);
WriteLn;
TextAttr := BVst;
Write(’Počet předmětů (max ’,MaxN,’): ’);
ReadLn(N);
WriteLn(’Jejich hmotnosti (desetinná čísla s tečkou): ’);
for K := 1 to N do
begin
X := 12*((K-1) mod 5)+2; Y := 7+((K-1) div 5);
GotoXY(X,Y);
Write(K,’: ’); ReadLn(M[K]);
end;
TextAttr := BVys;
Rozdil := 1000;
Rozloz(N);
Zaver(0);
Rozdil := 1000;
Zkusme(N);
Zaver(1);
TextAttr := BInf; GotoXY(2,22);
WriteLn(’Konec.’);
ReadLn
end. {program}

L i t e r a t u r a

[1] Trávníček, S.: Rozdělení napůl. Rozhledy matematicko-fyzikální, 87 (2012),
č. 3, str. 18–22.

